

Robotik Labor

Labor 2 “Bälle sortieren“

Baumann, Marc Matr. : 13855

Berroth, Kai-Uwe Matr. : 13842

Abgabe:

Unterschrift:

Versuch: Bälle sortieren

Unterschiedlich schwere Tennisbälle sollten mit Hilfe des IBM 7575 sortiert werden.

Hierfür stand ein Ballmagazin mit einem Kontaktschalter, 2 Waagen mit jeweils einem Kontaktschalter und drei Paletten zur Verfügung. Der Roboter ist mit einem Greifer ausgerüstet welcher sich öffnen und schließen lässt. Hiefür stehen die Befehle und `RELEASE ()`.

Lösung:

Um die Sortieraufgabe zu lösen wurde zuvor ein AML2 Programm im Simulator erstellt um die wesentlichen Funktionen zu testen. Die Waage wurde zur Gewichtsdetektierung folgendermaßen eingebunden: Zunächst wurde der Ball auf die empfindlichere Waage gelegt, löste er keinen Kontakt aus, wurde in ein Unterprogramm gesprungen um den Ball in die entsprechende Palette abzulegen. Bei Kontakt legte der Roboter greifer den Ball auf die Größere Waage um festzustellen ob es sich um einen schweren oder leichten Ball handelt.

Entsprechend wurde in eine Subroutine gesprungen.

Befinden sich in einer Palette bereits zwölf Bälle so wurde das Programm über eine Abfrage beendet.



Quelltext des AML2 Programms:

```
## Balle einordnen je nach Gewicht
## Marc Baumann und Kai-Uwe Berroth
#####
## Einlesen der 3 Definierpunkte pro Palette
## Palette 1 (leichte Baelle)
pul1 : NEW PT( -267.291, 290.524, 0., 0. );
pur1 : NEW PT( -443.931, 321.368, 0., 0. );
por1 : NEW PT( -487.033, 81.9304, 0., 0. );

## Palette 2 (mittlere Bälle)
pul2 : NEW PT( 46.765, 520.986, 0., 0.0072 );
pur2 : NEW PT( -127.765, 524.806, 0., 0.0036 );
por2 : NEW PT( -141.557, 291.039, 0., 0.0072 );

## Palette 3 (schwerer Bälle)
pul3 : NEW PT( 521.929, 171.038, 0., 0.0072 );
pur3 : NEW PT( 355.871, 181.941, 0., 0.0072 );
por3 : NEW PT( 338.693, -59.8474, 0., 0.0072 );
#####
#####
## Definiton der Palettenmatrizen
pal1 : NEW PALLET (pul1, pur1, por1, 3, 12);
pal2 : NEW PALLET (pul2, pur2, por2, 3, 12);
pal3 : NEW PALLET (pul3, pur3, por3, 3, 12);
#####
#####
## Position des Ballmagazines
magazin : NEW PT( 434.048, 292.159, 0., 43.9092 );
#####
#####
## Position der Ballwaagen
waage1 : NEW PT( 228.937, 372.957, 0., 42.7752 );
waage2 : NEW PT( 234.757, 447.179, 0., 42.7752 );
#####
#####
## Variablen
taste_a : NEW 1001;
kontaktmagazin : NEW 25;
kontaktwaage1 : NEW 26;
kontaktwaage2 : NEW 27;
a : NEW 0;
b : NEW 0;
c : NEW 0;
#####
#####
## Hauptprogramm
versuch2 : NEW SUBR();
#####
## Palettenzaehler auf 1 setzten
SETPART(pal1,1);
SETPART(pal2,1);
SETPART(pal3,1);
#####
SPEED =0.1;
#####
## Startbedingung wenn taste_a gedrueckt
PLED (1,TRUE);
PDISPLAY (1,'@n');
PDISPLAY (1, 'Start mit Taste A');
```

```

Schleife:      TESTI (Taste_a, False, Schleife);
               PLED (1,FALSE);
#####
#####
## magazin anfahren und gegf. warten oder ball holen
               Release ();
               Delay (1.5);
magazinanfahren: IF a EQ 12 THEN BRANCH (ausdiemaus);
                 IF b EQ 12 THEN BRANCH (ausdiemaus);
                 IF c EQ 12 THEN BRANCH (ausdiemaus);
                 SPEED =0.5;
                 PMOVE (magazin);
magazinwarten: TESTI (kontaktmagazin, False, magazinwarten);
                ZMOVE (-106.749);
                GRASP ();
                DELAY (0.5);
                ZMOVE (0);

#####
#####
## Ballwaegel anfahren und auswerten
SPEED = 0.2;
PMOVE (waage1);
ZMOVE (-15.6);
RELEASE();
DELAY(1.5);
TESTI (kontaktwaegel, FALSE, Palett1); ## wenn kein kontakt dann leicht
GRASP ();
DELAY (0.5);
ZMOVE(0);

#####
#####
## Ballwaage2 anfahren und auswerten
SPEED = 0.2;
PMOVE (waage2);
ZMOVE (-15.6);
RELEASE();
DELAY (1.5);
TESTI (kontaktwaage2, FALSE, Palette2);
BRANCH (Palette3);

#####
#####
## Palett1 leichte baelle
Palett1:      SPEED = 0.5;
               GRASP ();
               DELAY (0.5);
               ZMOVE (0);
               GETPART (pal1);    ## faehrt Palett1 an mit pos x
               ZMOVE (-92.5);
               RELEASE ();
               DELAY (1.5);
               ZMOVE (0);
               NEXTPART (pal1);   ## erhoehrt Palettenzaeler um 1
               a++;
               BRANCH (magazinanfahren);

#####
#####
## Palette2 mittlere baelle
Palette2:     SPEED = 0.5;
               GRASP ();
               DELAY (0.5);
               ZMOVE (0);
               GETPART (pal2);
               ZMOVE (-92.5);

```

```
        RELEASE ();
        DELAY (1.5);
        ZMOVE (0);
        NEXTPART (pal2);
        b++;
        BRANCH (magazinanfahren);
#####
#####
## Palette3 schwere baelle
Palette3:    SPEED = 0.5;
             GRASP ();
             DELAY (0.5);
             ZMOVE (0);
             GETPART (pal3);
             ZMOVE (-92.5);
             RELEASE ();
             DELAY (1.5);
             ZMOVE (0);
             NEXTPART (pal3);
             c++;
             BRANCH (magazinanfahren);
#####
ausdiemaus: PDISPLAY(1, '@N');
             PDISPLAY(1, 'Eine Palette voll!');
             HOME ();
             END;
```